



## UvA-DARE (Digital Academic Repository)

### Logic programming for knowledge-intensive interactive applications

Wielemaker, J.

**Publication date**  
2009

[Link to publication](#)

#### **Citation for published version (APA):**

Wielemaker, J. (2009). *Logic programming for knowledge-intensive interactive applications*. [Thesis, fully internal, Universiteit van Amsterdam].

#### **General rights**

It is not permitted to download or to forward/distribute the text or part of it without the consent of the author(s) and/or copyright holder(s), other than for strictly personal, individual use, unless the work is under an open content license (like Creative Commons).

#### **Disclaimer/Complaints regulations**

If you believe that digital publication of certain material infringes any of your rights or (privacy) interests, please let the Library know, stating your reasons. In case of a legitimate complaint, the Library will make the material inaccessible and/or remove it from the website. Please Ask the Library: <https://uba.uva.nl/en/contact>, or a letter to: Library of the University of Amsterdam, Secretariat, P.O. Box 19185, 1000 GD Amsterdam, The Netherlands. You will be contacted as soon as possible.

## Chapter 7

---

# SWI-Prolog and the Web

**About this chapter** This chapter is published in Theory and Practice of Logic Programming (Wielemaker et al. 2008). Written as a journal paper that provides an overview of using SWI-Prolog for web related tasks, it has some overlap with the previous chapters. The original paper has been extended with material from (Wielemaker et al. 2007): dispatching and session management in section 7.4.2 and the entire section 7.5.

Section 7.8 was contributed by the coauthor Zhisheng Huang and section 7.9 by Lourens van der Meij, both from the VU, Amsterdam.

**Abstract** Prolog is an excellent tool for representing and manipulating data written in formal languages as well as natural language. Its safe semantics and automatic memory management make it a prime candidate for programming robust Web services.

Where Prolog is commonly seen as a component in a Web application that is either embedded or communicates using a proprietary protocol, we propose an architecture where Prolog communicates to other components in a Web application using the standard HTTP protocol. By avoiding embedding in an external Web server, development and deployment become much easier. To support this architecture, in addition to the HTTP transfer protocol, we must support parsing, representing and generating the key Web document types such as HTML, XML and RDF.

This paper motivates the design decisions in the libraries and extensions to Prolog for handling Web documents and protocols. The design has been guided by the requirement to handle large documents efficiently.

The benefits of using Prolog for Web related tasks is illustrated using three case studies.

## 7.1 Introduction

The Web is an exciting place offering new opportunities to artificial intelligence, natural language processing and Logic Programming. Information extraction from the Web, reasoning in Web applications and the Semantic Web are just a few examples. We have deployed Prolog in Web related tasks over a long period. As most of the development on SWI-Prolog takes place in the context of projects that require new features, the system and its libraries provide extensive support for Web programming.

There are two views on deploying Prolog for Web related tasks. In the most commonly used view, Prolog acts as an embedded component in a general Web processing environment. In this role it generally provides reasoning tasks such as searching or configuration within constraints. Alternatively, Prolog itself can act as a stand-alone HTTP server as also proposed by ECLiPSe (Leth et al. 1996). In this view it is a component that can be part of any of the layers of the popular three-tier architecture for Web applications. Components generally exchange XML if used as part of the backend or middleware services and HTML if used in the presentation layer.

The latter view is in our vision more attractive. Using HTTP and XML over HTTP, the service is cleanly isolated using standard protocols rather than proprietary communication. Running as a stand-alone application, the interactive development nature of Prolog can be maintained much more easily than embedded in a C, C++, Java or C# application. Using HTTP, automatic testing of the Prolog components can be done using any Web oriented test framework. HTTP allows Prolog to be deployed in any part of the service architecture, including the realisation of complete Web applications in one or more Prolog processes.

When deploying Prolog in a Web application using HTTP, we must not only implement the HTTP transfer protocol, but also support parsing, representing and generating the important document types used on the Web, especially HTML, XML and RDF. Note that, being widely used open standards, supporting these document types is also valuable outside the context of Web applications.

This paper gives an overview of the Web infrastructure we have realised. Given the range of libraries and Prolog extensions that facilitate Web applications we cannot describe them in detail. Details on the library interfaces can be found in the manuals available from the SWI-Prolog Web site.<sup>1</sup> Details on the implementation are available in the source distribution. The aim of this paper is to give an overview of the required infrastructure to use Prolog for realising Web applications where we concentrate on scalability and performance. We describe our decisions for representing Web documents in Prolog and outline the interfaces provided by our libraries.

This paper illustrates the benefits of using Prolog for Web related tasks in three case studies: 1) SeRQL, an RDF query language for meta data management, retrieval and reasoning; 2) XDIG, an eXtended Description Logic interface, which provides ontology management and reasoning by processing DIG XML documents and communicating to external DL reasoners; and 3) A faceted browser on Semantic Web databases integrating meta-data from multiple

---

<sup>1</sup><http://www.swi-prolog.org>

collections of art-works. This case study serves as a complete Semantic Web application serving the end-user. Part II of this thesis contains two additional examples of applying the infrastructure described in this paper: PIDoc (chapter 8), an environment to support literate programming in Prolog and ClíoPatria (chapter 10), a web-server for thesaurus-based annotation and search.

This paper is organised as follows. Section 7.2 to section 7.3 describe reading, writing and representation of Web related documents. Section 7.4 describes our HTTP client and server libraries. Supporting AJAX and CSS, which form the basis of modern interactive web pages, is the subject of section 7.5. Section 7.6 describes extensions to the Prolog language that facilitate use in Web applications. Section 7.7 to section 7.9 describe the case studies.

## 7.2 XML and HTML documents

The core of the Web is formed by document standards and exchange protocols. This section discusses the processing of tree-structured documents transferred as SGML or XML. HTML, an SGML application, is the most commonly used document format on the Web. HTML represents documents as a tree using a fixed set of *elements* (tags), where the SGML DTD (Document Type Declaration) puts constraints on how elements can be nested. Each node in the hierarchy has a name (the element-name), a set of name-value pairs known as its attributes and *content*, a sequence of sub-elements and text (data).

XML is a rationalisation of SGML using the same tree-model, but removing many rarely used features as well as abbreviations that were introduced in SGML to make the markup easier to type and read by humans. XML documents are used to represent text using custom application-oriented tags as well as a serialisation format for arbitrary data exchange between computers. XHTML is HTML based on XML rather than SGML.

In this section we discuss parsing, representing and generating SGML/XML documents. Finally (section 7.2.3) we compare our work with PiLLow (Gras and Hermenegildo 2001).

### 7.2.1 Parsing and representing XML and HTML documents

The first SGML parser for SWI-Prolog was created by Anjo Anjewierden based on the SP parser.<sup>2</sup> A stable Prolog term-representation for SGML/XML trees plays a similar role as the DOM (*Document Object Model*) representation in use in the object-oriented world. The term-structure we use is described in figure 7.1.

Below, we motivate some of the key aspects of the representation of figure 7.1.

- Representation of text by a Prolog atom is biased by the use of SWI-Prolog which has no length-limit on atoms and atoms that can represent UNICODE text as motivated in section 7.6.2. At the same time SWI-Prolog stacks are limited to 128MB each on 32-bit machines. Using atoms, only the structure of the tree is represented on the stack while the bulk of the data is stored on the unlimited heap. Using lists of character

---

<sup>2</sup><http://www.jclark.com/sp/>

```

<document>      ::= list-of <content>
<content>       ::= <element> | <pi> | <cdata> | <sdata> | <ndata>
<element>       ::= element(<tag>, list-of <attribute>, list-of <content>)
<attribute>     ::= <name> = <value>
<pi>            ::= pi(<atom>)
<sdata>         ::= sdata(<atom>)
<ndata>         ::= ndata(<atom>)
<cdata>, <name> ::= <atom>
<value>         ::= <svalue> | list-of <svalue>
<svalue>        ::= <atom> | <number>

```

Figure 7.1: SGML/XML tree representation in Prolog. The notation list-of  $\langle x \rangle$  describes a Prolog list of terms of type  $\langle x \rangle$ .

codes is another possibility adopted by both PiLLOW (Gras and Hermenegildo 2001) and ECLiPse (Leth et al. 1996). Two observations make lists less attractive: lists use two stack cells per character while practical experience shows text is frequently processed as a unit only. For (HTML) text-documents we profit from the compact representation of atoms. For XML documents representing serialised data-structures we profit from frequent repetition of the same value represented by a single handle to a shared atom.

- The value of an attribute that is declared in the DTD as multi-valued (e.g., NAMES) is returned as a Prolog list. This implies the DTD must be available to get unambiguous results. With SGML this is always true, but not with XML. Attribute-values are always returned as a single atom if no type information is provided by the DTD.
- Optionally, attribute-values of type NUMBER or NUMBERS are mapped to Prolog a prolog number or list of Prolog numbers. In addition to the DTD issues mentioned above, this conversion also suffers from possible loss of information. Leading zeros and different floating point number notations used are lost after conversion. Prolog systems with bounded arithmetic may also not be able to represent all values. Still, automatic conversion is useful in many applications, especially those involving serialised data-structures.
- Attribute values are represented as *Name=Value*. Using *Name(Value)* is an alternative. The *Name=Value* representation was chosen for its similarity to the SGML notation and because it avoids the need for univ (= . .) for processing argument-lists.

**Implementation** The SWI-Prolog SGML/XML parser is implemented as a C-library that has been built from scratch to create a lightweight parser. Total source is 11,835 lines.

The parser provides two interfaces. Most natural to Prolog is `load_structure(+Src, -DOM, +Options)` which parses a Prolog stream into a term as described above. Alternatively, `sgml_parse/2` provides an *event-based* parser making call-backs on Prolog for the SGML *events*. The call-back mode can process unbounded documents in streaming mode. It can be mixed with the term-creation mode, where the handler for *begin* calls the parser to create a term-representation for the content of the element. This feature is used to process large files with a repetitive record structure in limited memory. Section 7.3.1 describes how this is used to process RDF documents.

When we decided to implement a lightweight parser the only alternative available was the SP<sup>3</sup> system. Lack of flexibility of the API and the size of SP (15× larger than ours) caused us to implement our own parser. Currently, there are other lightweight XML and HTML libraries available, some of which may satisfy our requirements.

Full documentation is available from the SWI-Prolog website.<sup>4</sup> The SWI-Prolog SGML parser has been adopted by XSB Prolog.

## 7.2.2 Generating Web documents

There are many approaches to generating Web pages from programs in general and Prolog in particular. Below we present some use-cases, requirements and alternatives that must be considered.

- How much of the document is generated from dynamic data and how much is static? Pages that are static except for a few strings are best generated from a template using variable substitution. Pages that consist of a table generated from dynamic data are best entirely generated from the program.
- For program generated pages we can choose between direct printing and generating using a language-native syntax (Prolog), for example `format('<b>bold</b>')` or `print_html(b(bold))`. The second approach can guarantee well-formed output, but requires the programmer to learn the mapping between Prolog syntax and HTML syntax. Direct printing requires hardly any knowledge beyond the HTML syntax.
- Documents that contain a significant static part are best represented in the markup language where special constructs insert program-generated parts. A popular approach implemented by PHP<sup>5</sup> and ASP<sup>6</sup> is to add a reserved element such as `<script>` or use the SGML/XML *programming instruction* written as `<?...?>`. The obvious name PSP (Prolog Server Pages) is in use by various projects taking this approach.<sup>7</sup> An-

<sup>3</sup><http://www.jclark.com/sp/>

<sup>4</sup><http://www.swi-prolog.org/packages/sgml2pl.html>

<sup>5</sup>[www.php.net](http://www.php.net)

<sup>6</sup>[www.microsoft.com](http://www.microsoft.com)

<sup>7</sup><http://www.prologonlinereference.org/psp.psp>,

<http://www.benjaminjohnston.com.au/template.prolog?t=psp>,

[http://www.ifcomputer.com/inap/inap2001/program/inap\\_bartenstein.ps](http://www.ifcomputer.com/inap/inap2001/program/inap_bartenstein.ps)

other approach is PWP<sup>8</sup> (Prolog Well-formed Pages). It is based on the principle that the input is well-formed XML that interacts with Prolog through additional attributes. Output is guaranteed to be well-formed XML. Because we did not yet encounter a real need for any of these approaches in projects, our current infrastructure does not include any of them.

- Page *transformation* is realised by parsing the original document into its tree representation, managing the tree and writing a new document from the tree. Managing the source-text directly is not reliable because due to alternative character encodings, entity usage and different use of SGML abbreviations there are many different source-texts that represent the same tree. The `load_structure/3` predicate described in section 7.2 together with output primitives from the library `sgml_write.pl` provide this functionality. The XDIG case study described in section 7.8 follows this approach.

### 7.2.2.1 Generating documents using DCG

The traditional method for creating Web documents is using print routines such as `write/1` or `format/2`. Although simple and easily explained to novices, the approach has serious drawbacks from a software engineering point of view. In particular the user is responsible for HTML quoting, character encoding issues and proper nesting of HTML elements. Automated validation is virtually impossible using this approach.

Alternatively, we can produce a DOM term as described in section 7.2 and use the library `sgml_write.pl` to create the HTML or XML document. Such documents are guaranteed to use proper nesting of elements, escape sequences and character encoding. The terms however are big, deeply nested and hard to read and write. Prolog allows them to be built from skeletons containing variables. In our opinion, the result is not optimal due to the unnatural order of statements and the introduction of potentially many extra variables as illustrated in figure 7.2. In this figure we first generate a small table using `mkthumbnail/3`, which is then inserted at the right location into the skeleton page using the variable `ThumbNail`. As the number of partial DOM structures that must be created grows, this style of programming quickly becomes unreadable.

We decided to design a two-step process. The first step is formed by the DCG rule `html//1`,<sup>9</sup> which translates a Prolog term into a list of high-level HTML/XML commands that are handed to `html_print/1` to realise proper quoting, character encoding and layout. The intermediate format is of no concern to the user. Generated from a Prolog term with the same nesting as the target HTML document, consistent opening and closing of elements is guaranteed. In addition to variable substitution which is provided by Prolog we allow calling rules. Rules are invoked by a term `\Rule` embedded in the argument of `html//1`. Figure 7.3 illustrates our approach, producing the same document as figure 7.2 in a more

<sup>8</sup><http://www.cs.otago.ac.nz/staffpriv/ok/pwp.pl>

<sup>9</sup>The notation `<name>/(arity)` refers to the grammar rule `<name>` with the given `<arity>`, and consequently the predicate `<name>` with arity `<arity>+2`.

```

...
mkthumbnail(URL, Caption, Thumbnail),
output_html([ h1('Photo gallery'),
              Thumbnail
            ]).

```

```

mkthumbnail(URL, Caption, Term) :-
  Term = table([ tr(td([halign=center],
                      img([src=URL],[ ])),
                tr(td([halign=center],
                      Caption))
              ])

```

Figure 7.2: Building a complex DOM tree from smaller components by using Prolog variables (*Thumbnail* in this figure).

readable fashion. Any reusable part of the page generation can be translated into a DCG rule. Using the *Rule* syntax it is clear which parts of the argument of `html//1` is directly translated into HTML elements and which part is expanded in a rule.

```

...
html([ h1('Photo gallery'),
       \thumbnail(URL, Caption)
     ]).

```

```

thumbnail(URL, Caption) -->
  html(table([ tr(td([halign=center], img([src=URL],[ ])),
                tr(td([halign=center], Caption))
              ])).

```

Figure 7.3: Defining reusable fragments (`thumbnail//2`) using library `html_write.pl`

In our current implementation rules are called using meta-calling from `html//1`. Using `term_expansion//2` it is straightforward to move the rule invocation out of the term, using variable substitution similar to `PiLLow`. It is also possible to recursively expand the generated tree and validate it to the HTML DTD at compile-time and even insert omitted tags at compile-time to generate valid XHMTL from an incomplete specification. An overview of the argument to `html//1` is given in figure 7.4.

```

<html>      ::= list-of <content> | <content>
<content>  ::= <atom>
            | <tag>(list-of <attribute>, <html>)
            | <tag>(<html>)
            | \<rule>
<attribute> ::= <name>(<value>)
<tag>, <entity> ::= <atom>
<value>    ::= <atom> | <number>
<rule>     ::= <callable>

```

Figure 7.4: The `html//1` argument specification

### 7.2.3 Comparison with PiLLOW

The PiLLOW library (Gras and Hermenegildo 2001) is a well established framework for Web programming based on Prolog. PiLLOW defines `html2terms/2`, converting between an HTML string and a document represented as a Herbrand term. There are fundamental differences between PiLLOW and the primitives described here.

- PiLLOW creates an HTML document from a Herbrand term that is passed to `html2terms/2`. Complex terms are composed of partial terms passed as Prolog variables, a technique we described in the second paragraph of section 7.2.2.1. Frequently used HTML constructs are supported using reserved terms using dedicated processing. This realises a form of macro expansion using a predefined and fixed set of macros. We use DCGs and the `\Rule` construct, which makes it evident which terms directly refer to HTML elements and which function as a macro. In addition, the user can define application-specific reusable fragments in a uniform way.
- The PiLLOW parser does not create the SGML document tree. It does not insert omitted tags, default attributes, etcetera. As a result, HTML documents that differ only in omitted tags and whether or not default attributes are included in the source, produce different terms. In our approach the term representation is equivalent, regardless of the input document. This is illustrated in figure 7.5. Having a canonical DOM representation greatly simplifies processing parsed HTML documents.

## 7.3 RDF documents

Where the datamodel of both HTML and XML is a tree-structure with attributes, the datamodel of the Semantic Web (SW) language RDF<sup>10</sup> consists of *{Subject, Predicate, Object} triples*. Both *Subject* and *Predicate* are URIs.<sup>11</sup> *Object* is either a URI or a *Literal*. As the

<sup>10</sup><http://www.w3.org/RDF/>

<sup>11</sup>URI: *Uniform Resource Identifier* is like a URL, but need not refer to an existing resource on the Web.

```
[env(table, [], [tr$[], td$[], "Hello"])]
```

```
[element(table, [],
  [ element(tbody, [],
    [ element(tr, [],
      [ element(td, [ rowspan='1',
                    colspan='1'
                  ],
                ['Hello' ])]))]
  )]
```

Figure 7.5: Term representations for `<table><tr><td>Hello</td></tr></table>` in PiLLow (top) and our parser (bottom). Our parser completes the `tr` and `td` environments, inserts the omitted `tbody` element and inserts the defaults for the `rowspan` and `colspan` attributes

*Object* of one triple can be the *Subject* of another, a set of triples forms a graph, where each edge is labelled with a URI (the *Predicate*) and each vertex is either a URI or a literal. Literals have no out-going edges. Figure 7.6 illustrates this.




Figure 7.6: Sample RDF graph. Ellipses are vertices representing URIs. Quoted text is a literal. Edges are labelled with URIs.

A number of languages are layered on top of the RDF triple model. RDFS provides a frame-based representation. The OWL-dialects<sup>12</sup> provide three increasingly complex Web ontology languages. SWRL<sup>13</sup> is a proposal for a rule language. The W3C standard for exchanging these triple models is an XML application known as RDF/XML.

As there are multiple XML tree representations for the same triple-set, RDF documents cannot be processed at the level of the XML-DOM as described in section 7.2. A triple-

<sup>12</sup><http://www.w3.org/2004/owl/>

<sup>13</sup><http://www.w3.org/Submission/SWRL/>

or graph-based structure is the most natural choice for representing an RDF document in Prolog. The nodes of this graph are formed by URIs and literals. Because a URI is a string and the only operation defined on URIs by SW languages is testing equivalence, a Prolog atom is the most obvious candidate to represent a URI. One may consider using a term  $\langle namespace \rangle : \langle localname \rangle$ , but given that decomposing a URI into its namespace and localname is only relevant during I/O we consider this an inferior choice. Writing the often long URIs as a quoted atom in Prolog source code harms readability and complicates changing namespaces. Therefore, the RDF library comes with a compile-time rewrite mechanism based on `goal_expansion/2` that allows for writing resources in Prolog source as  $\langle ns \rangle : \langle local \rangle$ . Literals are expressed as `literal(Value)`, where *Value* represents the value of the literal. The full Prolog representation of RDF elements is given in figure 7.7.

$\langle subject \rangle$ , $\langle predicate \rangle$	::=	$\langle \text{URI} \rangle$
$\langle object \rangle$	::=	$\langle \text{URI} \rangle$
		<code>literal(<math>\langle lit\_value \rangle</math>)</code>
$\langle lit\_value \rangle$	::=	$\langle text \rangle$
		<code>lang(<math>\langle langid \rangle</math>, <math>\langle text \rangle</math>)</code>
		<code>type(<math>\langle \text{URI} \rangle</math>, <math>\langle text \rangle</math>)</code>
$\langle \text{URI} \rangle$ , $\langle text \rangle$	::=	$\langle atom \rangle$
$\langle langid \rangle$	::=	$\langle atom \rangle$ (ISO639)

Figure 7.7: RDF types in Prolog.

The typical SW use-scenario is to ‘harvest’ triples from multiple sources and collect them in a database before reasoning with them. Prolog can represent data as a Herbrand term on the stack or as predicates in the database. Given the relatively static nature of the RDF data as well as desired access from multiple threads, using the Prolog database is the most obvious choice. Here we have two options. One is the predicate `rdf(Subject, Predicate, Object)` using the argument types described above. The alternative is to map each RDF predicate on a Prolog predicate `Predicate(Subject, Object)`. We have chosen for `rdf/3` because it supports queries with uninstantiated predicates better and a single predicate is easier to manage than an unbounded set of predicates with unknown names.

### 7.3.1 Input and output of RDF documents

The RDF/XML parser is realised as a Prolog library on top of the XML parser described in section 7.2. Similar to the XML parser it has two interfaces. The predicate `load_rdf(+Src, -Triples, +Options)` parses a document and returns a Prolog list of `rdf(S,P,O)` triples. Note that despite harvesting to the database is the typical use-case scenario, the parser delivers a list of triples for maximal flexibility. The predicate `process_rdf(+Src, :Action, +Options)` exploits the mixed call-back/convert mode of the XML parser to process the RDF file one *description* (record) at a time, calling *Action* with a list of triples extracted from the description. Figure 7.8 illustrates how this is used by the storage module to load unbounded

files with limited stack usage. Source location as `<file>:<line>` is passed to the `Src` argument of `assert_triples/2`.

```
load_triples(File, Options) :-
    process_rdf(File, assert_triples, Options).

assert_triples([], _).
assert_triples([rdf(S,P,O)|T], Src) :-
    rdf_assert(S, P, O, Src),
    assert_triples(T, Src).
```

Figure 7.8: Loading triples using `process_rdf/3`

In addition to named URIs, RDF resources can be *blank-nodes*. A blank-node (short *bnode*) is an anonymous resource that is created from an in-lined description. Figure 7.9 describes the dimensions of a painting as a compound instance of class *Dimension* with width and height properties. The *Dimension* instance has no URI. Our parser generates an identifier that starts with a double underscore, followed by the source and a number. The double underscore is used to identify bnodes. Source and number are needed to guarantee the bnode is unique.

```
<Painting rdf:about="...">
  <dimension>
    <Dimension width="45" height="50"/>
  </dimension>
</Painting>
```

Figure 7.9: Blank node to express the compound dimension property

The parser from XML to RDF triples covers the full RDF specification, including UNICODE handling, RDF datatypes and RDF language tags. The Prolog source is 1,788 lines. It processes approximately 9,000 triples per second on an AMD 1600+ based computer. Implementation details and evaluation of the parser are described in chapter 3 (Wielemaker et al. 2003b).

We have two libraries for writing RDF/XML. One, `rdf_write_xml(+Stream, +Triples)`, provides the inverse of `load_rdf/2`, writing an XML document from a list of `rdf(S,P,O)` terms. The other, called `rdf_save/2` is part of the RDF storage module described in section 7.3.2 and writes a database directly to a file or stream. The first

(`rdf_write_xml/2`) is used to exchange computed graphs to external programs using network communication, while the second (`rdf_save/2`) is used to save modified graphs back to file. The resulting code duplication is unfortunate, but unavoidable. Creating a temporary graph in a database requires potentially much memory, and harms concurrency, while graphs fetched from the database into a list may not fit in the Prolog stacks and is also considerably slower than a direct write.

### 7.3.2 Storing and indexing RDF triples

If we collect RDF triples in a database we must provide an API to query the RDF graph. Obviously, the natural primary API to query an RDF graph is a pure non-deterministic predicate `rdf(?S,?P,?O)`. Considering that RDF graphs tend to be large (see below), indexing the RDF database is crucial for good performance. Table 7.1 illustrates the calling pattern from a real-world application counting 4 million triples. When designing the implementation `rdf(?S,?P,?O)` we can exploit the restricted set of Prolog datatypes used in RDF graphs as described by figure 7.7. The RDF store was developed in the context of projects (see section 9.3) which formulated the following requirements.

- Upto at least 10 million triples on 32-bit hardware.<sup>14</sup>
- Fast graph traversal using any instantiation pattern.
- Case-insensitive search on literals.
- Prefix search on literals for completion in the User Interface.
- Searching for words that appear in literals.
- Multi-threaded access allowing for concurrent readers.
- Transaction management and persistent store.
- Maintain source information, so we can update, save or remove data based on its source.
- Fast load/save of current state.

Our first version of the database used the Prolog database with secondary tables to improve indexing. As requirements pushed us against the limits of what is achievable in a 32-bit address-space we decided to implement the low level store in C. Profiting from the known uniform structure of the data we realised about two times more compact storage with better indexing than using a pure Prolog approach. We took the following design decisions for the C-based storage module:

- The RDF *predicates* are represented as unique entities and organised according to the `rdfs:subPropertyOf` relation in multiple hierarchies. Each cloud of connected properties is equipped with a reachability matrix. See section 3.4.1.1 for details.

<sup>14</sup>our current aim is 300 million on 64-bit with 64 Gb memory

Index pattern			Calls
-	-	-	58
+	-	-	253,554
-	+	-	62
+	+	-	23,292,353
-	-	+	633,733
-	+	+	7,807,846
+	+	+	26,969,003

Table 7.1: Call-statistics on a real-world system

- Literals are kept in an AVL tree, sorted case-insensitive and case-preserving (e.g., AaBb... ). Numeric literals precede all non-numeric and are kept sorted on their numeric value. Storing literals in a separate sorted table avoids the need to store duplicates and allows for indexed search for prefixes and numeric values. It also allows for monitoring creation and destruction of literals to maintain derived tables such as stemming or double metaphone (Philips 2000) based on `rdf_monitor/3` described below. The space overhead of maintaining the table is roughly cancelled by avoiding duplicates. Experience on real data ranges between -5% and +10%.
- Resources are represented by Prolog atom-handles. The hash is computed from the handle-value. Note that avoiding the translation between Prolog atom and text avoids both duplication of data and table-lookup. We consider this a crucial aspect.
- Each triple is represented by the atom-handle for the subject, predicate-pointer, atom-handle or literal pointer for object, a pointer to the source, a line number, a general bit-flag field and 6 'hash-next' pointers covering all indexing patterns except for +,+,+ and +,-,+ . Queries using the pattern +,-,+ are rare. Fully instantiated queries internally use the pattern +,+,-, assuming few values on the same property. Considering experience with real data we will probably add a +,+,+ index in the future. The un-indexed table is a simple linked list. The others are hash-tables that are automatically resized if they become too populated.

The store itself does not allow for writes while there are active reads in progress. If another thread is reading, the write operation will stall until all threads have finished reading. If the thread itself has an open choicepoint a permission error exception is raised. To arrive at meaningful update semantics we introduced *transactions*. The thread starting a transaction obtains a write-lock, initially allowing readers to proceed. During the transaction all changes are recorded in a linked list of actions. If the transaction is ready for commit, the thread denies access to new readers and waits for all readers to vanish before updating the database. Transactions are realised by `rdf_transaction(:Goal)`. If *Goal* succeeds, its choicepoints are discarded and the transaction is committed. If *Goal* fails or raises an excep-

tion the transaction is discarded and `rdf_transaction/1` returns failure or exception. Transactions can be nested. Nesting a transaction places a transaction-mark in the list of actions of the current transaction. Committing implies removing this mark from the list. Discarding removes all action cells following the mark as well as the mark itself.

It is possible to monitor the database using `rdf_monitor(:Goal, +Events)`. Whenever one of the monitored events happens *Goal* is called. Modifying actions inside a transaction are called during the commit. Modifications by the monitors are collected in a new transaction which is committed immediately after completing the preceding commit. Monitor events are `assert`, `retract`, `update`, `new_literal`, `old_literal`, transaction begin/end and file-load. *Goal* is called in the modifying thread. As this thread is holding the database write lock, all invocations of monitor calls are fully serialised.

Although the 9,000 triples per second of the RDF/XML parser ranks it among the fast parsers, loading 10 million triples takes nearly 20 minutes. For this reason we developed a binary format. The format is described in chapter 3 (Wielemaker et al. 2003b) and loads approximately 20 times faster than RDF/XML, while using about the same space. The format is independent from byte-order and word-length, supporting both 32- and 64-bit hardware.

Persistency is achieved through the library `rdf_persistency.pl`, which uses `rdf_monitor/3` to maintain a set of files in a directory. Each source known to the database is represented by two files, one file representing the initial state using the quick-load binary format and one file containing Prolog terms representing changes, called the *journal*.

### 7.3.3 Reasoning with RDF documents

We have identified two approaches for defining an API that supports reasoning based on Semantic Web languages such as RDFS and OWL. Both languages define an *abstract syntax* that defines their semantics in terms of conceptual entities on which the language is based. Both languages are also defined by triples that can be deduced from the set of explicitly provided triples, the *deductive closure* of the explicitly provided triples under the language. The extra triples that can be derived are called *entailed* triples. Each of these views on the language can be used to define an API:

- *Abstract syntax based API*

The abstract syntax introduces concepts that form the basis of the language, such as *class*, *individual* or *restriction*. If this approach is applied to RDFS, the API provides predicates such as `rdfs_individual_of(?Resource, ?Class)` and `rdfs_subclass_of(?Sub, ?Super)`. The SWI-Prolog library `rdfs.pl` and the ClioPatria (chapter 10) module `owl.pl` provide an API based on the abstract syntax.

- *Entailment reasoning based API*

Semantic web query languages such as `serql` provide access to the deductions by querying the full deductive closure instead of only the explicitly provided RDF state-

ments. Because the deductive closure is also a set of triples, this technique requires no additional API.

We use this technique in our implementation of the `serql` (and `sparql`) query languages, as illustrated in figure 7.19. In this scenario the module `rdfs` exports an alternative definition of `rdf/3` that is true for any triple in the deductive closure. The implementation uses backward reasoning.

Figure 7.10 illustrates the two approaches. *Triples* lists the explicit triples. Both APIs return the explicit fact that *mary* is a *woman* as well as the derived (using the semantics of RDFS) fact that *mary* is a *human*. Both interfaces provide the same semantics. Both interfaces can be implemented using backward reasoning or forward reasoning. If the entailment interface is implemented as a backward reasoner, it needs to use different rulesets depending on the RDF predicate (second argument of the triple). Implemented as a forward reasoner, the derived triples are added to the database. This simplifies querying, but makes it harder to have multiple reasoners available to the application programmer at the same time. An API based on the abstract syntax and using backward reasoning can easily allow for multiple reasoners in the same application and makes it explicit what type of reasoning is used. This adds to the conceptual clarity of programs that use this API. An API based on entailment reasoning can switch between reasoners for the whole application without any change to the application code.

<i>Triples</i>	<i>Entailment API</i>	<i>Abstract syntax API</i>
mary type woman .	?- <b>rdf</b> (mary, type, X).	?- <b>rdfs_individual_of</b> (mary, X).
woman type Class .	X = woman ;	X = woman ;
woman subclassOf human .	X = human ;	X = human ;
human type Class .	false.	false.

Figure 7.10: Different APIs for RDFS

## 7.4 Supporting HTTP

HTTP, or HyperText Transfer Protocol, is the key W3C standard protocol for exchanging Web documents. All browsers and Web servers implement it. The initial version of the protocol was simple. The client request consists of a single line of the format *<action> <path>*, the server replies with the requested document and closes the connection. Version 1.1 of the protocol is more complicated, providing additional name-value pairs in the request as well as the reply, features to request status such as modification time, transfer partial documents, etcetera.

Adding HTTP support in Prolog, we must consider both the client- and server-side. In both cases our choice is between doing it in Prolog or re-using an existing application or

library by providing an interface for it. We compare our work with PiLLow (Gras and Hermenegildo 2001) and the ECLiPSe HTTP services (Leth et al. 1996).

Given a basic TCP/IP socket library, writing an HTTP client is trivial (our client counts just 505 lines of code). Both PiLLow and ECLiPSe include a client written in Prolog. The choice between embedding Prolog in an existing server framework and providing a pure Prolog-based server implementation is more complicated:

- The server is much more complex, which implies there is more to gain by re-using external code. Initially, the core server library counted 1,784 lines; the current server architecture counts over 9,000 lines.
- A single computer can only host one server at port 80 used by default for public HTTP. Using an alternate port for middleware and storage tier components is no problem, but use as a public server often conflicts with firewall or proxy settings. This can be solved using a proxy server such as the Apache *mod\_proxy*.<sup>15</sup>
- Servers almost by definition introduce security risks. Administrators are reluctant to see non-proven software in the role of a public server. Using a proxy as above also reduces this risk because it blocks (some) malformed requests.

Despite these observations, we consider, like the ECLiPSe team, a pure Prolog-based server worthwhile. As argued in section 7.6.1, many Prolog Web applications profit from using state stored in the server. Large resources such as WordNet (Miller 1995) cause long startup times. In such cases the use of CGI (Common Gateway Interface) is not appropriate as a new copy of the application is started for each request. PiLLow resolves this issue by using *Active Modules*, where a small CGI application talks to a continuously running Prolog server using a private protocol. Using a Prolog HTTP server and optionally a proxy has the same benefits, but based on a standard protocol, it is much more flexible.

Another approach is embedding Prolog in another server framework such as the Java-based Tomcat server. Although feasible, embedding non-Java-based Prolog systems in Java is complicated. Embedding through *jni* introduces platform and Java version dependent problems. Connecting Prolog and Java concurrency models and garbage collection is difficult and the resulting system is much harder to manage by the user than a pure Prolog-based application.

In the following sections we describe our HTTP client and server libraries. An overall overview of the modules and their dependencies is given in figure 7.11. The modules in this figure are described in the subsequent sections.

### 7.4.1 HTTP client libraries

We support two clients. The first (`http_open.pl`) is a lightweight client that only supports the HTTP GET method by means of `http_open(+URL, -Stream, +Options)`. *Options*

<sup>15</sup>[http://httpd.apache.org/docs/1.3/mod/mod\\_proxy.html](http://httpd.apache.org/docs/1.3/mod/mod_proxy.html)




Figure 7.11: Module dependencies of the HTTP library. The modules are described in section 7.4.1 and section 7.4.2.

allows for setting a timeout or proxy as well as getting information from the reply-header such as the size of the document. The `http_open/3` predicate internally handles HTTP 3XX (redirect) replies. Other not-ok replies are mapped to a Prolog exception. After reading the document the user must close the returned stream-handle using the standard Prolog `close/1` predicate. This predicate makes accessing an HTTP resource as simple as accessing a local file. The second library, called `http_client.pl`, provides support for HTTP POST and a plugin interface that allows for installing handlers for documents of specified MIME-types. It shares `http_header.pl` with the server libraries for DCG-based creation and parsing of HTTP headers. Currently provided plugins include `http_mime_plugin.pl` to handle *multipart* MIME messages and `http_sgml_plugin.pl` for automatically parsing HTML, XML and SGML documents. Figure 7.12 shows the code for fetching a URL and parsing the returned HTML document it into a Prolog term as described in section 7.2.

Both the `PILLOW` and `ECLIPSE` approach return the document's content as a string. Using an intermediate string is often a waste of memory resources and limits the maximum size of documents that can be processed. In contrast, our interface is stream-based (`http_open/3`). The `http_get/3` and `http_post/4` interfaces allows for plugin-based processing of the input stream. Stream-based processing avoids potentially large intermediate data structures and allows for processing unbounded documents.

```

?- use_module(library('http/http_client')).
?- use_module(library('http/http_sgml_plugin')).

?- http_get('http://www.swi-prolog.org/', DOM, []).
DOM = [ element(html,
              [ version = '-//W3C//DTD HTML 4.0 Transitional//EN'
              ],
              [ element(head, [],
                        [ element(title, [],
                                  [ 'SWI-Prolog\'s Home']), ...

```

Figure 7.12: Fetching an HTML document

### 7.4.2 The HTTP server library

Both to simplify re-use of application code and to make it possible to use the server without committing to a large infrastructure we adopted the reply-strategy of the CGI protocol, where the handler writes a page consisting of an HTTP header followed by the document content. Figure 7.13 provides a simple example that returns the request-data to the client. By importing `thread_http.pl` we implicitly selected the multi-threaded server model. Other models provided are `inetd_http`, causing the (Unix) `inet` daemon to start a server for each request and `xpce_http` which uses I/O multiplexing realising multiple clients without using Prolog threads. The logic of handling a single HTTP request given a predicate realising the handler, an input and output stream is implemented by `http_wrapper`.

```

:- use_module(
    library('http/thread_http')).

start_server(Port) :-
    http_server(reply, [port(Port)]).

reply(Request) :-
    format('Content-type: text/plain\n',
          writeln(Request)).

```




Figure 7.13: A simple HTTP server. The right window shows the client and the format of the parsed request.

Replies other than “200 OK” are generated using a Prolog exception. Recognised replies are defined by the predicate `http_reply(+Reply, +Stream, +Header)`. For example to indicate that the user has no access to a page we must use the following call.

```
throw(http_reply(forbidden(URL))).
```

Failure of the handler raises a “404 existence error” reply, while exceptions other than the ones described above raise a “500 Server error” reply.

### 7.4.2.1 The HTTP dispatching code

The core HTTP library handles all requests through a single predicate specified by `http_server/2`. Normally this predicate is defined ‘multifile’ to split the source of the server over multiple files. This approach proved inadequate for a larger server with multiple developers for the following reasons:

- There is no way to distinguish between non-existence of an HTTP location and failure of the predicate due to a programming error. This is an omission in itself, but with a larger project and multiple developers it becomes more serious.
- There is no easy way to tell where the specific clause is that handles an HTTP location.
- As the order of clauses in a multi-file predicate that come from different files is ill defined, it is not easy to reliably redefine the service behind a given HTTP location. Redefinition is desirable for re-use as well as for experiments during development.

To overcome these limitations we introduced a new library `http_dispatch.pl` that defines the directive `:- http_handler(Location, Predicate, Options)`. The directive is handled by `term_expansion/2` and is mapped to a multi-file predicate. This predicate in turn is used to build a Prolog term stored in a global variable that provides fast search for locations. Modifications to the multi-file predicate cause recomputation of the Prolog term on the next HTTP request. *Options* can be used to specify access rights, a priority to allow overruling existing definitions and assignment of the request to a new thread (*spawn*). Typically, each location is handled by a dedicated predicate. Based on the handler definitions, we can easily distinguish failure from non-existence as well as find, edit and debug the predicate implementing an HTTP location.

### 7.4.2.2 Form parameters

The library `http_parameters.pl` defines `http_parameters(+Request, ?Parameters)` to fetch and type-check parameters transparently for both GET and POST requests. Figure 7.14 illustrates the functionality. Parameter values are returned as atoms. If large documents are transferred using a POST request the user may wish to revert to `http_read_data(+Request, -Data, +Options)` underlying `http_get/3` to process arguments using plugins.

```

reply(Request) :-
    http_parameters(Request,
                    [ title(Title, [optional(true)]),
                      name(Name,   [length >= 2]),
                      age(Age,    [integer])
                    ]), ...

```

Figure 7.14: Fetching HTTP form data

### 7.4.2.3 Session management

The library `http_session.pl` provides session management over the otherwise stateless HTTP protocol. It does so by adding a cookie using a randomly generated code if no valid session id is found in the current request. The interface to the user consists of a predicate to set options (timeout, cookie-name and path) and a set of wrappers around `assert/1` and `retract/1`, the most important of which are `http_session_assert(+Data)`, `http_session_retract(?Data)` and `http_session_data(?Data)`. In the current version the data associated with sessions that have timed out is simply discarded. Session-data does not survive the server.

Note that a session generally consists of a number of HTTP requests and replies. Each *request* is scheduled over the available worker threads and requests belonging to the same session are therefore normally not handled by the same thread. This implies no session state can be stored in global variables or in the control-structure of a thread. If such style of programming is wanted the user must create a thread that represents the session and setup communication from the HTTP-worker thread to the session thread. Figure 7.15 illustrates the idea.

### 7.4.2.4 Evaluation

The presented server infrastructure is currently used by many internal and external projects. Coding a server is similar to writing CGI handlers and running in the interactive Prolog process is much easier to debug. As Prolog is capable of reloading source files in the running system, handlers can be updated while the server is running. Handlers running during the update are likely to die on an exception though. We plan to resolve this issue by introducing read/write locks. The protocol overhead of the multi-threaded server is illustrated in table 7.2.

## 7.5 Supporting AJAX: JSON and CSS

Recent web-technology is moving towards extensive use of CSS (Cascading Style Sheets) and JavaScript. The use of JavaScript has evolved from short code snippets adding visual

```

reply(Request) :-
    % HTTP worker
    ( http_session_data(thread(Thread))
    -> true
    ; thread_create(session_loop([], Thread,
        [detached(true)]),
        http_session_assert(thread(Thread))
    ),
    current_output(CGIOut),
    thread_self(Me),
    thread_send_message(Thread,
        handle(Request, Me, CGIOut)),
    thread_get_message(_Done).

```

```

session_loop(State) :-
    % Session thread
    thread_get_message(handle(Request, Sender, CGIOut)),
    next_state(Request, State, NewState, CGIOut).
    thread_send_message(Sender, done).

```

Figure 7.15: Managing a session in a thread. The `reply/1` predicate is part of the HTTP worker pool, while `session_loop/1` is executed in the thread handling the session. We omitted error handling for readability of the example.

Connection	Elapsed	Server CPU	Client CPU
Close	20.84	11.70	7.48
Keep-Alive	16.23	8.69	6.73

Table 7.2: HTTP performance executing a trivial query 10,000 times. Times are in seconds. Localhost, dual AMD 1600+ running SuSE Linux 10.0

effects through small libraries for presenting menus, etc. to extensive *widget* libraries such as YUI<sup>16</sup>. Considering the classical three-tier web application model (storage, application logic and presentation), the presentation tier is gradually migrated from the server towards the client. Typical examples are Google Maps and Google Calendar. Modern web *applications* consist of JavaScript libraries with some glue code that accesses one or more web-servers through an API that provides the content represented using the JSON serialisation format.<sup>17</sup>

This is an attractive development for deploying Prolog as a web-server. Although we

<sup>16</sup><http://developer.yahoo.com/yui/>

<sup>17</sup><http://www.json.org/>

have seen that Prolog can be deployed in any of the three tiers, its support for the presentation layer is weak due to the lack of ready-to-use resources. If the presentation layer is moved to the client we can reuse the presentation resources from the JavaScript community.

An AJAX-based web application is an HTML page that links in JavaScript widget libraries, other JavaScript web applications, application specific libraries and glue code as well as a number of CSS files, some of which belong to the loaded libraries while others provide the customisation. This design requires special attention in two areas. First, using a naive approach with declarations that load the required resources is hard to maintain and breaks our DCG-based reusability model because the requirements must be specified in the HTML *<head>* element. Second, most AJAX libraries use JSON (JavaScript Object Notation) as serialisation format for data. The subsequent two sections describe how we deal with each of these.

### 7.5.1 Producing HTML head material

Modern HTML+CSS and AJAX-based web-pages require links to scripts and style files in the HTML *<head>* element. XHTML documents that use RDFa (Adida and Birbeck 2007) or other embedded XML that requires additional namespaces, require XML namespace declarations, preferably also in the HTML *<head>*. One of the advantages of HTML generation as described is that pages can be composed in a modular way by calling the HTML-generating DCG rules. This modularity can only be maintained if the DCG rule produces both the HTML and causes the HTML *<head>* to be expanded with the CSS, JavaScript and XML namespaces on which the generated HTML depends.

We resolved this issue by introducing a *mail* system into the HTML generator. The grammar rule `html_receive(+MailBox, :Goal)` opens a mailbox at the place in the HTML token sequence where it appears. The grammar rule `html_post(+MailBox, +Content)` posts HTML content embedded with calls to other grammar rules to *MailBox*. A post processing phase collects all posts to a mailbox into a list and runs *Goal* on list to produce the final set of HTML tokens. This can be used for a variety of tasks where rules need to emit HTML at certain places in the document. Figure 7.16 gives an example dealing with footnotes at the bottom of a page.

The mail system is also used by the library `html_head.pl` to deal with JavaScript and CSS resources that are needed by an HTML page. The library provides a mailbox called `head`. The directive `html_resource(+Resource, +Attributes)` allows for making declarations on dependencies between resources such as JavaScript and CSS files. The rule `html_requires(+Resource)` specifies that we need a particular CSS or JavaScript file, as well as all other resource files that are needed by that file according to the `html_resource/2` declarations. The `html_receive//2` emit wrapper that is inserted by `html_head.pl` performs the following steps:

1. Remove all duplicates from the requested resources.
2. Add implied requirements that are not yet requested.

```

:- use_module(library(http/html_write)).

footnote_demo :-
    reply_html_page(
        title('Footnote demo'),
        body([ \generate_body,
              \html_receive(footnotes, emit_footnotes)
            ])).

:- dynamic current_fn/1.

generate_body -->
    { assert(current_fn(1) },
      html([ h1('Footnote demo'),
            p([ 'We use JavaScript',
              \footnote('Despite the name, JavaScript is \
                        essentially unrelated to the Java \
                        programming language. '), ' for ...'
            ])
          ])).

footnote(Content) -->
    { retract(current_fn(I)),
      I2 is I + 1,
      assert(current_fn(I2))
    },
    html(sup(class(footnote_ref), a(href('#fn'+I2), I))),
    html_post(footnotes, \emit_footnote(I, Content)).

%%      emit_footnotes(+List)// is det.
%
%      Emit the footnotes, Called delayed from html_receive/2.

emit_footnotes([]) -->
    [].                                     % no footnotes
emit_footnotes(List) -->
    html(div(class(footnotes),
            List)).

emit_footnote(I, Content) -->
    html(div(class(footnote),
            [ span(class(footnote_index),
                  a(name(fn+I), I)),
              \html(Content)
            ])).

```

Figure 7.16: Using the HTML mail system to generate footnotes

3. Try to apply *aggregate* scripts (see below).
4. Do a topological sort based on the dependencies, placing all requirements before the scripts that require them.
5. Emit HTML code for the sorted final set of resources.

Step 3 was inspired by the YUI framework which, in addition to the core JavaScript files, provides a number of *aggregate* files that combine a number of associated script files into a single file without redundant white space and comments to reduce HTTP traffic and enhance loading performance. If more than half of the scripts in a defined aggregate are required, the scripts are replaced by the aggregate. The 50% proportion is suggested in the YUI documentation, but not validated. The current implementation uses manually created and declared aggregates. The library has all information to create and maintain aggregates automatically based on usage patterns. This would provide maximal freedom distributing JavaScript and CSS over files for maximal modularity and conceptual clarity with optimal performance.

Figure 7.17 shows a rule from ClioPatria (chapter 10, Wielemaker et al. 2008). The rule is a reusable definition for displaying a single result found by the ClioPatria search engine. Visualisation of a single result is defined in CSS. Calling this rule generates the required HTML, and at the same time ensures that the CSS declarations (`path.css`) are loaded from the HTML `<head>`. This mechanism provides simple reusability of rules that produce HTML fragments together with the CSS and JavaScript resources that provide the styling and interactive behaviour.

```
result_with_path(Result, Path) -->
    html_requires(css('path.css')),
    html(span(class(result),
              [ Result,
                span(class(path), \html_path(Path))
              ])).
```

Figure 7.17: A single rules produces HTML and ensures that the required CSS is loaded.

## 7.5.2 Representing and converting between JSON and Prolog

JSON is a simple format that defines *objects* as a list of name/value pairs, arrays as an unbounded sequence of values and a few primitive datatypes: strings, numbers, boolean (`true` and `false`) and the constant `null`. For example, a point in a two-dimensional plane can be represented in JSON as `{"x":10, "y":20}`. If explicit typing is desired, one might use `{"type":"point", "x":10, "y":20}`. The JSON syntax is a subset of the SWI-Prolog syntax,<sup>18</sup> but the resulting term is very unnatural from Prolog's perspective: attribute

<sup>18</sup>Not of the ISO Prolog syntax because JSON is UNICODE (UTF-8) based and allows for the `\uXXXX` syntax to express UNICODE code points.

names map to strings instead of atoms, wasting space and making lookup unnecessarily slow. The `{...}` notation is for Prolog's grammar rules and unnatural to process. Instead, the natural Prolog representation of a point object is a term `point(10,20)`.

This section describes how JSON can be integrated into Prolog. To facilitate JSON handling from Prolog we created two libraries. The first, `json.pl` reads and writes JSON terms from/to a Prolog stream in an unambiguous format that covers the full JSON specification. The generic format is defined as in figure 7.18. For example, the point object above is represented as `json([x=10,y=20])`.

Object	::=	json([])   json([Pair, ...])
Pair	::=	Name = Value
Name	::=	<i>&lt;atom&gt;</i>
Array	::=	[]   [Value, ...]
Value	::=	<i>&lt;atom&gt;</i>   <i>&lt;number&gt;</i>   Object   Array
		@(true)   @(false)   @(null)

Figure 7.18: Unambiguous representation of a JSON term, using the same BNF structure as the JSON specification.

The second, `json_convert.pl` supports type checking and conversion to a 'natural' Prolog term. Complex types are defined by the name/arity of a Prolog compound term. Primitive types are `atom`, `integer`, etc. JSON interface types are declared using the directive `json_object(+Specification)`, which we introduce through the example below. The example defines two types. The first specifies the type `person/2` with arguments `name` and `address`. In turn, `address` is defined by the street, house-number and city, all of which are covered by Prolog primitive types.

```
:- json_object
    person(name:atom, address:address),
    address(street:atom, number:integer, city:atom).
```

With these declarations, `prolog_to_json(+Prolog, -JSON)` converts a term such as `person(mary, address(kerkstraat, 42, amsterdam))` into the unambiguous JSON representation `json([mary, json([kerkstraat, 42, amsterdam])]`. The inverse operation is performed by `json_to_prolog(+JSON, -Prolog)`. The library enforces the type declarations.

## 7.6 Enabling extensions to the Prolog language

SWI-Prolog has been developed in the context of projects, many of which focused on managing Web documents and protocols. In the previous sections we have described our Web enabling libraries. In this section we describe extensions to the ISO-Prolog standard (Deransart et al. 1996) we consider crucial for scalable and comfortable deployment of Prolog as an agent in a Web centred world.

### 7.6.1 Multi-threading

Concurrency is necessary for applications for the following reasons:

- Network delays may cause communication of a single transaction to take long. It is not acceptable if such incidents block access for other clients. This can be achieved using multiplexed I/O, multiple processes handling requests in a pool or multiple threads.
- CPU-intensive services must be able to deploy multiple CPUs. This can be achieved using multiple instances of the service and load-balancing or a single server running on multi-processor hardware or a combination of the two.

As indicated, none of the requirements above require multi-threading support in Prolog. Nevertheless, we added multi-threading (chapter 6, Wielemaker 2003a) because it resolves delays and exploiting multi-CPU hardware for medium-scale applications while greatly simplifying deployment and debugging in a platform independent way. A multi-threaded server also allows maintaining state for a specific session or even state shared between multiple sessions simply in the Prolog database. The advantages of this are described in (Szeredi et al. 1996), using the or-parallel Aurora to serve multiple clients. This is particularly interesting for accessing the RDF database described in section 7.3.2.

### 7.6.2 Atoms and UNICODE support

UNICODE<sup>19</sup> is a character encoding system that assigns unique integers (code-points) to all characters of almost all scripts known in the world. In UNICODE 4.0, the code-points range from 1 to 0x10FFFF. UNICODE can handle documents in different scripts (e.g., Latin and Hebrew) as well as documents that contain text from multiple scripts. This feature greatly simplifies applications that must be able to deal with multiple scripts, such as web applications serving a world-wide audience. Traditional HTML applications commonly insert special symbols through entities such as the copyright (©) sign, Greek and mathematical symbols, etcetera. Using UNICODE we can represent all entity values uniformly as plain text. UTF-8, an encoding of UNICODE as a sequence of bytes, is at the heart of XML and the Semantic Web.

HTML documents can be represented using Prolog strings because Prolog integers can represent all UNICODE code-points. As we have claimed in section 7.2 however, using Prolog strings is not the most obvious choice. XML attribute names and values can contain arbitrary UNICODE characters, which requires the unnatural use of strings for these as well. If we consider RDF, IRIs can have arbitrary UNICODE characters<sup>20</sup> and we want to represent IRIs as atoms to exploit compact storage as well as fast equivalence testing. Without UNICODE support in atoms we would have to encode UNICODE in the atom using escape sequences. All this patchwork can be avoided if we demand the properties below for Prolog atoms.

---

<sup>19</sup><http://www.Unicode.org/>

<sup>20</sup><http://www.w3.org/TR/rdf-concepts/#section-Graph-URIref>

- Atoms represent text in UNICODE
- Atoms have no limit on their length
- The Prolog implementation allows for a large number of atoms, both to represent URIs and to represent text in HTML/XML documents. SWI-Prolog's maximum number of atoms is  $2^{25}$  (32 million) on 32-bit hardware.
- Continuously running servers cannot allow memory leaks and therefore processing dynamic data using atoms requires atom garbage collection.

## 7.7 Case study — A Semantic Web Query Language

In this case-study we describe the SWI-Prolog SeRQL implementation, currently distributed as an integral part of ClíoPatria chapter 10 (Wielemaker et al. 2008). SeRQL is an RDF query language developed as part of the Sesame project<sup>21</sup> (Broekstra et al. 2002). SeRQL uses HTTP as its access protocol. Sesame consists of an implementation of the server as a Java servlet and a Java client-library. By implementing a compatible framework we made our Prolog-based RDF storage and reasoning engine available to Java clients. The Prolog SeRQL implementation uses all of the described SWI-Prolog infrastructure and building it has contributed significantly to the development of the infrastructure. Figure 7.19 lists the main components of the server.




Figure 7.19: Module dependencies of the SeRQL system. Arrows denote ‘imports from’ relations.

The *entailment* modules are plugins that implement the entailment approach to RDF reasoning described in section 7.3.3. They implement `rdf/3` as a pure predicate, adding implicit triples to the raw triples loaded from RDF/XML documents. Figure 7.20 shows the somewhat simplified entailment module for RDF. The negations avoid duplicate results. This

<sup>21</sup><http://www.openrdf.org>

is not strictly necessary, but if the test is cheap we expect that the extra cost will generally be returned in subsequent processing steps. The multifile rule `serql:entailment/2` registers the module as entailment module for the `serql` system. New modules can be loaded dynamically into the platform, providing support for other SW languages or application-specific server-side reasoning.

```
:- module(rdf_entailment, [rdf/3]).

rdf(S, P, O) :-
    rdf_db:rdf(S, P, O).
rdf(S, rdf:type, rdf:'Property') :-
    rdf_db:rdf(_, S, _),
    \+ rdf_db:rdf(S, rdf:type, rdf:'Property').
rdf(S, rdf:type, rdfs:'Resource') :-
    rdf_db:rdf_subject(S),
    \+ rdf_db:rdf(S, rdf:type, rdfs:'Resource').

:- multifile serql:entailment/2.

serql:entailment(rdf, rdf_entailment).
```

Figure 7.20: RDF entailment module

The `serql` parser is a DCG-based parser that translates a `serql` query text into a compound goal calling `rdf/3` and predicates from the `serql` runtime library which provide comparison and functions built into the `serql` language. The resulting control-structure is passed to the query optimiser chapter 4 (Wielemaker 2005) which uses statistics maintained by the RDF database to reorder the pure `rdf/3` calls for best performance. The optimiser uses a generate-and-evaluate approach to find the optimal order. Considering the frequently long conjunctions of `rdf/3` calls, the conjunction is split into independent parts. Figure 7.21 illustrates this in a simple example.

HTTP access consists of two parts. The human-centred portal consists of HTML pages with forms to manage the server as well as view statistics, load and unload documents and run `serql` queries interactively presenting the result as an HTML table. Dynamic pages are generated using the `html_write.pl` library described in section 7.2.2.1. Static pages are served from HTML files by the Prolog server. Machines use HTTP POST requests to provide query data and get a reply in XML or RDF/XML.

The system knows about various RDF input and output formats. To reach modularity the kernel exchanges RDF graphs as lists of terms `rdf(S,P,O)` and result-tables as lists of terms using the functor `row` and arity equal to the number of columns in the table. The system calls a multifile predicate using the format identifier and data to produce the results in the

```

...
rdf(Paper, author, Author),
rdf(Author, name, Name),
rdf(Author, affiliation, Affil),
...

```

Figure 7.21: Split rdf conjunctions. After executing the first `rdF/3` query *Author* is bound and the two subsequent queries become independent. This is also true for other orderings, so we only need to evaluate 3 alternatives instead of 3! (6).

requested output format. The HTML output format uses `html_write.pl`. The RDF/XML format uses `rdF_write_xml/2` described in section 7.3.1. Both `rdF_write_xml/2` and the other XML output format use straight calls `format/3` to write the document, where quoting values is realised by quoting primitives provided by the SGML/XML parser described in section 7.2. Using direct writing instead of techniques described in section 7.2.2.1 avoids potentially large intermediate datastructures and is not complicated given the often simple structure of the documents.

## Evaluation

Development of the `seRQL` server and the SWI-Prolog web libraries is too closely integrated to use it as an evaluation of the functionality provided by the Web enabling libraries. We compared our server to Sesame, written in Java. The source code of the Prolog-based server is 6,700 lines, compared to 86,000 for Sesame. As both systems have very different coverage in functionality and can re-use libraries at different levels it is hard to judge these figures. Both answer trivial queries in approximately 5ms on a dual AMD 1600+ PC running Linux 2.6. On complex queries the two systems perform very differently. Sesame's forward reasoning makes it handle some RDFS queries much faster. Sesame does not contain a query optimiser which causes order-dependent and sometimes very long response times on conjunctions.

The power of LP where programs can be handled as data is exploited by parsing the `seRQL` query into a program and optimising this program by manipulating it as data, after which we can simply call it to answer the query. The non-deterministic nature of `rdF/3` allows for a trivial translation of the query to a non-deterministic program that produces the answers on backtracking.

The server only depends on the standard SWI-Prolog distribution and therefore runs unmodified on all systems supporting SWI-Prolog. It has been tested on Windows, Linux and MacOS X.

All infrastructure described is used in the server. We use `format/3`, exploiting XML quoting primitives provided by the Prolog XML library to print highly repetitive XML files

such as the `seRQL` result-table. Alternatively we could have created the corresponding DOM term and call `xml_write/2` from the library `sgml_write.pl`.

## 7.8 Case study — XDIG

In section 7.7 we have discussed the case study how SWI-Prolog is used for a RDF query system, i.e., a meta-data management and reasoning system. In this section we describe a Prolog-powered system for ontology management and reasoning based on Description Logics (DL). DL has greatly influenced the design of the W3C ontology language OWL. The DL community, called DIG (DL Implementation Group) have developed a standard for accessing DL reasoning engines called the DIG description logic interface<sup>22</sup> (Bechhofer et al. 2003), DIG interface for short. Many DL reasoners like Racer (Haarslev and Möller 2001) and FACT (Horrocks 1999) support the DIG interface, allowing for the construction of highly portable and reusable DL components or extensions.

In this case study, we describe XDIG, an eXtended DIG Description Logic interface, which has been implemented on top of the SWI-Prolog Web libraries. XDIG is a platform that provides a DIG *proxy* in which application specific transformations and extensions can be realised. The DIG interface uses an XML-based messaging protocol on top of HTTP. Clients of a DL reasoner communicate by means of HTTP POST requests. The body of the request is an XML encoded message which corresponds to the DL concept language. Where OWL is based on the triple model described in section 7.3, DIG statements are grounded directly in XML. Figure 7.22 shows a DIG statement which defines the concept *MadCow* as a cow which eats brains, part of sheep.

```
<equalc>
  <catom name='mad+cow' />
  <and>
    <catom name='cow' />
    <some>
      <ratom name='eats' />
      <and>
        <catom name='brain' />
        <some>
          <ratom name='part+of' />
          <catom name='sheep' />
        </some>
      </and>
    </some>
  </and>
</equalc>
```

Figure 7.22: a DIG statement on MadCow

<sup>22</sup><http://dl.kr.org/dig/>

### 7.8.1 Architecture of XDIG

The XDIG libraries form a framework to build DL reasoners that have additional reasoning capabilities. XDIG serves as a regular DL reasoner via its corresponding DIG interface. An intermediate XDIG server can make systems independent from application specific characteristics. A highly decoupled infrastructure significantly improves the reusability and applicability of software components.




Figure 7.23: Architecture of XDIG

The general architecture of XDIG is shown in figure 7.23. It consists of the following components:

**XDIG Server** The XDIG server deals with requests from ontology applications. It supports our extended DIG interface: in addition to the standard DIG/DL requests, like 'tell' and 'ask' it supports requests such as changing system settings. The library `dig_server.pl` implements the XDIG protocol on top of the Prolog HTTP server described in section 7.4.2. The predicate `dig_server(+Request)` is called from the HTTP server to process a client's *Request* as illustrated in figure 7.13. XDIG server developers have to define the predicate `my_dig_server_processing(+Data, -Answer, +Options)`, where *Data* is the parsed DIG XML requests and *Answer* is term `answer(-Header, -Reply)`. *Reply* is the XML-DOM term representing the answer to the query.

**DIG Client** XDIG is designed to rely on an external DL reasoner. It implements a regular DIG interface client and calls the external DL reasoner to access the standard DL reasoning capabilities. The predicate `dig_post(+Data, -Reply, +Options)` posts the data to the external DIG server. The predicates are defined in terms of the predicate `http.post/4` and others in the HTTP and XML libraries.

**Main Control Component** The library `dig_process.pl` provides facilities to analyse DIG statements such as finding concepts, individuals and roles, but also decide on *satisfiability* of concepts and consistency. Some of this processing is done by analysing

the XML-DOM representation of DIG statements in the local repository, while satisfiability and consistency checking is achieved by accessing external DL reasoners through the DIG client module.

**Ontology Repository** The Ontology Repository serves as an internal knowledge base (KB), which is used to store multiple ontologies locally. These ontology statements are used for further processing when the reasoner receives an ‘ask’ request. The main control component usually selects parts from the ontologies to post them to an external DL reasoner and obtain the corresponding answers. This internal KB is also used to store system settings.

As DIG statements are XML-based, XDIG stores statements in the local repository using the XML-DOM representation described in section 7.2. The tree model of XDIG data has been proved to be convenient for DIG data management.

Figure 7.24 shows a piece of code from the XDIG defining the predicate `direct_concept_relevant(+DOM, ?Concept)` which checks if a set of *Statements* is directly relevant to a *Concept*, namely the *Concept* appears in the body of a statement in the list. The predicate `direct_concept_relevant/2` has been used to develop PION for reasoning with inconsistent ontologies, and DION for inconsistent ontology debugging.

```

direct_concept_relevant(element(catom, Atts, _), Concept) :-
    memberchk(name=Concept, Atts).
direct_concept_relevant(element(_, _, Content), Concept) :-
    direct_concept_relevant(Content, Concept).
direct_concept_relevant([H|T], Concept) :-
    (   direct_concept_relevant(H, Concept)
    ;   direct_concept_relevant(T, Concept)
    ).

```

Figure 7.24: `direct_concept_relevant` checks that a concept is referenced by a DIG statement

## 7.8.2 Application

XDIG has been used to develop several DL reasoning services. PION is a reasoning system that deals with inconsistent ontologies<sup>23</sup> (Huang and Visser 2004; Huang et al. 2005), MORE is a reasoner<sup>24</sup> (Huang and Stuckenschmidt 2005) that supports multiple versions of the same ontology and DION is a debugger of inconsistent ontologies<sup>25</sup> (Schlobach and Huang 2005).

<sup>23</sup><http://wasp.cs.vu.nl/sekt/pion>

<sup>24</sup><http://wasp.cs.vu.nl/sekt/more>

<sup>25</sup><http://wasp.cs.vu.nl/sekt/dion>

With the support of an external DL reasoner like Racer (Haarslev and Möller 2001), DION can serve as an inconsistent ontology debugger using a bottom-up approach.

## 7.9 Case study — Faceted browser on Semantic Web database integrating multiple collections

In this case study we describe a pilot for the STITCH-project<sup>26</sup> that aims at finding solutions for the problem of integrating controlled vocabularies such as thesauri and classification systems in the Cultural Heritage domain. The pilot consists of the integration of two collections – the Medieval Illuminations of the Dutch National Library (Koninklijke Bibliotheek) and the Masterpieces collection from the Rijksmuseum – and development of a user interface for browsing the merged collections. One requirement within the pilot is to use “standard Semantic Web techniques” during all stages, so as to be able to evaluate their added value. An explicit research goal was to evaluate existing “ontology mapping” tools. The problem could be split into three main tasks:

- Gathering data, i.e., collecting records of the collections and controlled vocabularies they use and transforming these into RDF.
- Establishing semantic links between the vocabularies using off-the-shelf ontology mapping tools.
- Building a prototype User Interface (UI) to access (search and browse) the integrated collections and experiment with different ways to access them using a Web server.

SWI-Prolog has been used to realise all three tasks. To illustrate our use of the SWI-Prolog Web libraries we focus on their application in the prototype user interface because it is the largest subsystem using these libraries.

### 7.9.1 Multi-Faceted Browser

Multi-Faceted Browsing is a search and browse paradigm where a collection is accessed by refining multiple (preferably) structured aspects (called facets) of its elements. For the user interface and user interaction we have been influenced by the approach of Flamenco (Hearst et al. 2002). The Multi-Faceted Browser is implemented in SWI-Prolog. All data is stored in an RDF database, which can be either an external *SeRQL* repository or an in-memory SWI-Prolog RDF database. The system consists of three components, *RDF-interaction*, which deals with RDF-database storage and access, *HTML-code generation*, for the creation of Web pages and the *Web server* component, implementing the HTTP server. They are discussed in the following sections.

---

<sup>26</sup><http://stitch.cs.vu.nl>

### 7.9.1.1 RDF-interaction

We first describe the content of the RDF database before explaining how to access it. The RDF database contains:

- 750 records from the Rijksmuseum, and 1000 from the Koninklijke Bibliotheek.
- An RDF representation of hierarchically structured *facets*. Hierarchical structuring is defined using SKOS<sup>27</sup>, an RDF schema to represent controlled vocabularies.
- Mappings between SKOS Concept Schemes used in the different collections.
- Portal-specific information as “*Site Configuration Objects*” (SCO), identified by URIs with properties defining what collections are part of the setup, what facets are shown, and also values for the constant text in the Web page presentation and other User Interface configuration properties. Multiple SCOs may be defined in a repository.

The in-memory RDF store contains, depending on the number of mappings and structured vocabularies that are stored in the database, about 300,000 RDF triples. The Sesame store contains more triples (520,000) as its RDFS-entailment implementation implies generation of derived triples (see section 7.7).

**RDF database access** Querying the RDF store for compound results such as a list of artworks where each artwork is enriched with information on how it must be displayed (e.g., title, thumbnail, collection) based on HTTP query arguments consists of three steps: 1) building `serql` queries from the HTTP query arguments, 2) passing them on to the `serql`-engine, gathering the result rows and 3) finally post-processing the output, e.g., counting elements and sorting them. Figure 7.25 shows an example of a generated `serql` query. Finding matching records involves finding records annotated by the facet value or by a value that is a hierarchical descendant of facet value. We implemented this by interpreting records as instances of SKOS concepts and using the transitive and reflexive properties of the `rdfs:subClassOf` property. This explains for example `{Rec} rdf:type {<http://www.telin.nl/rdf/topia#Paintings>}` in figure 7.25.

The `serql`-query interface contains timing and debugging facilities for single queries; for flexibility it provides access to an external `serql` server<sup>28</sup> for which we used Sesame<sup>29</sup>, but also to the in-memory store of the SWI-Prolog `serql` implementation described in section 7.7.

<sup>27</sup><http://www.w3.org/2004/02/skos/>

<sup>28</sup>We used the `sesame_client.pl` library that provides an interface to external `serql` servers, packaged with the SWI-Prolog `serql` library

<sup>29</sup><http://www.openrdf.org>

```

SELECT  Rec, RecTitle, RecThumb, CollSpec
FROM    {SiteId} rdfs:label {"ARIATOPS-NONE"};
        mfs:collection-spec {CollSpec} mfs:record-type {RT};
        mfs:shorttitle-prop {TitleProp};
        mfs:thumbnail-prop {ThumbProp},
{Rec}   rdf:type {RT}; TitleProp {RecTitle};
        ThumbProp {RecThumb},
{Rec}   rdf:type {<http://www.telin.nl/rdf/topia#AnimalPieces>},
{Rec}   rdf:type {<http://www.telin.nl/rdf/topia#Paintings>}
USING NAMESPACE skos = <http://www.w3.org/2004/02/skos/core#>,
mfs = <http://www.cs.vu.nl/STITCH/pp/mf-schema#><br>

```

Figure 7.25: An example of a serQL query, which returns details of records matching two facet values (AnimalPieces and Paintings)

### 7.9.1.2 HTML-code generation

We used the SWI-Prolog `html_write.pl` library described in section 7.2.2.1 for our HTML-code generation. There are three distinct kinds of Web pages the multi-faceted browser generates, the portal access page, the refinement page and the single collection-item page. The DCG approach to generating HTML code made it easy to share HTML-code generating procedures such as common headers and HTML code for refinement of choices. The HTML-code generation component contains some 140 DCG rules (1200 lines of Prolog code of which 800 lines are DCG rules), part of which are simple list-traversing rules such as the example of Figure 7.26.

### 7.9.1.3 Web Server

The Web server is implemented using the HTTP server library described in section 7.4.2. The Web server component itself is small. It follows the skeleton code described in Figure 7.13. In our case the `reply/1` predicate extracts the URL root and parameters from the URL. The *Site Configuration Object*, which is introduced in section 7.9.1.1, is returned by the RDF-interaction component based on the URL root. It is passed on to the HTML-code generation component which generates Web content.

## 7.9.2 Evaluation

This case study shows that SWI-Prolog is effective for building applications in the context of the Semantic Web. In a single month a fully functional prototype portal has been created providing structured access to multiple collections. The independence of any external libraries and the full support of all libraries on different platforms made it easy to develop and install in different operating systems. All case study software has been tested to install and run transparently both on Linux and on Microsoft Windows.

```

objectstr([],_O, _Cols,_Args) --> [].
objectstr([RowObjs|ObjectsList], Offset, Cols, Args) -->
    { Perc is 100/Cols },
    html(tr(valign(top),
            \objectstd(RowObjs, Offset, Perc, Args))),
    { Offset1 is Offset + Cols },
    objectstr(ObjectsList, Offset1, Cols, Args).

objectstd([],_,_,_) --> [].
objectstd([Url|RowObjects], Index, Percentage, Args) -->
    {
        ..
        construct_href_index(..., HRef),
        missing_picture_txt(Url, MP)
    },
    html(td(width(Percentage),
            a(href(HRef),img([src(Url),alt(MP)])))),
    { Index1 is Index + 1 },
    objectstd(RowObjects, Index1, Percentage, Args).

```

Figure 7.26: Part of the html code generation for displaying all the images of a query result in an HTML table

At the start of the pilot project we briefly evaluated existing environments for creating multi-faceted browsing portals: We considered the software available from the Flamenco Project (Hearst et al. 2002) and the OntoViews Semantic Portal Creation Tool (Mäkelä et al. 2004). The Flamenco software would require developing a translation from RDF to the Flamenco data representation. OntoViews heavily uses Semantic Web techniques, but the software was unnecessarily complex for our pilot, requiring a number of external libraries. This together with our need for flexible experiments with various setups made us decide to build our own prototype.

The prototype allowed us to easily experiment with and develop various interesting ways of providing users access to integrated heterogeneous collections (van Gendt et al. 2006).

## 7.10 Conclusion

We have presented an overview of the libraries and Prolog language extensions we have implemented and which we provide to the Prolog community as Open Source resources. The following components have been identified as vital for using Prolog in (semantic) web related tasks:

- *HTTP client and server support*

Instead of using proprietary protocols between Prolog and other components in a web server architecture, we propose the use of the web HTTP protocol. Based on HTTP, Prolog can be deployed anywhere in the server architecture.

- *Web document support*

In addition to the transfer protocol, standardisation of document formats is what makes the web work. We propose a representation of XML/SGML, HTML, RDF and JSON documents as Prolog terms. Standardisation of these Prolog terms is a first requirement to enable portable libraries processing web documents. We presented a modular and extensible mechanism to generate HTML documents. The infrastructure automatically includes CSS and JavaScript on which the document depends.

- *RDF storage and querying*

RDF is naturally represented by a Prolog predicate `rdf(Subject, Predicate, Object)`. We implemented `rdf/3` as a foreign extension to Prolog because this allows us to optimise the representation by exploiting the restrictions and use patterns provided by the Semantic Web languages.

Proper support of the above components is not possible without extending Prolog with at least the following features:

- *Threading*

Availability of multi-CPU hardware and requirements for responsive and scalable web services demands concurrency. This is particularly true when representing large amounts of knowledge as Prolog rules. Multiple threads provide scalable reasoning with this knowledge without duplication.

- *Atom handling*

Atoms are the obvious representation for IRIS in RDF as well as element and attribute names in XML. This requires UNICODE atoms as well and large numbers of atoms. In addition, atoms can be used to represent attribute values and (textual) content in XML, which requires atoms of unbounded length and atom garbage collection.

Considering the three tier model for implementing web services, the middleware, dealing with the application logic, is the most obvious tier for exploiting Prolog. Flexibility provided by using HTTP for communication does not limit Prolog to the middleware. In the case studies presented in this paper we have seen Prolog active as storage component (section 7.7), middleware (section 7.8) and in the presentation tier (section 7.9). In chapter 10, the Prolog server (ClioPatria) initially handled all three tiers. Later, part of the presentation was moved to JavaScript running in the web-browser to improve dynamic behaviour and exploit reusable JavaScript libraries. ClioPatria uses all infrastructure described in this chapter.

Figure 7.27 shows the ‘Semantic Web layer cake’ that illustrates the architecture of the Semantic Web and which parts of the cake are covered by the infrastructure described in this chapter.




Figure 7.27: The Semantic Web layer cake by Tim Burners Lee. The dark grey boxes are fully covered. For light grey areas, ontologies are covered by standard languages but our coverage of the standard is partial while the other layers are not covered by clear standards, but Prolog is a promising language to explore them.

Development in the near future is expected to concentrate on Semantic Web reasoning, such as the translation of SWRL rules to logic programs. Such translations will benefit from tabling to realise more predictable response-times and allow for more declarative programs. We plan to add more support for OWL reasoning, possibly supporting vital relations for ontology mapping such as `owl:sameAs` in the low-level store. We also plan to add PSP or PWP-like (see section 7.2.2) page-generation facilities.

## Acknowledgements

Preparing this paper as well as maturing RDF and HTTP support was supported by the MultimediaN<sup>30</sup> project funded through the BSIK programme of the Dutch Government. Other projects that provided a major contribution to this research include the European projects IMAT (ESPRIT 29175) and GRASP (AD 1008) and the ICES-KIS project “Multimedia Information Analysis” funded by the Dutch government. The XDIG case-study is supported by FP6 Network of Excellence European project SEKT (IST-506826). The faceted browser case-study is funded by CATCH, a program of the Dutch Organisation NWO. We acknowledge the SWI-Prolog community for feedback on the presented facilities, in particular Richard O’Keefe for his extensive feedback on SGML support.

<sup>30</sup>[www.multimedien.nl](http://www.multimedien.nl)